

Math 255 - Computer Organization

Parameter Passing on the PDP-8 : Parameter Slots

The words following the **jms I Div** instruction are used to hold parameters ("parameter slots"). In the example below a subroutine to divide A by B is called. The dividend (A) is loaded into the accumulator and the divisor (B) is inserted into the parameter slot following the subroutine call

```
Again,  tad B          / store divisor in
        dca Divide+1   /   parameter slot
        tad A          / load dividend
Divide, jms i Div      / divide A by B
        0              / parameter slot holds divisor
        dca Q          / Store quotient
        tad Divide+1   / check remainder (parameter slot)
        dca R          / Store remainder
```

The subroutine extracts the divisor from the parameter slot. Just before it returns, it increments the contents of Xdiv to jump over the parameter slot when it returns.

```
/
/ Subroutine to Divide using repeated subtraction - Page 29
/
*7200
XDiv,  0
        dca xdividend / store dividend parameter from accumulator
        tad i xdiv     / load divisor parameter from parameter slot
        cia           / negate it and
        dca xdivisor   / store in divisor
        dca xquotient  / zero out quotient
        tad xdividend  / load dividend
        skp           / skip over quotient increment
        isz xquotient  / increment quotient if subtraction successful
        tad xdivisor   / subtract divisor from dividend
        sma           / test if negative
        jmp .-3       / loop
        tad i xdiv     / restore remainder
        dca i xdiv     / store remainder in parameter slot
        tad xquotient  / accumulator gets quotient
        isz xdiv       / increment return address to skip parameter
        jmp i xdiv     / slot and return
xdividend, 0          / local dividend
xdivisor,  0          / local divisor
xquotient, 0          / local quotient
```

If additional parameter slots are used, XDiv must be incremented to obtain the second parameter. In the example below both the dividend and divisor are passed via parameter slots.

```
*7200
XDiv,  0
        cla cll
        tad i XDiv     / load dividend
        dca xdividend  / store locally
        isz XDiv       / increment XDiv
        tad i Xdiv     / load divisor
        dca xdivisor   / store locally
```